



# 競技プログラミングの鉄則\*

やさしい理系お兄ちゃん

February 1, 2024

## Contents

0 はじめに	2
1 アルゴリズムと計算量	2
1.1 導入問題	2
1.2 全探索 1	3
1.3 全探索 2	4
1.4 2進法	5
1.5 チャレンジ問題	6
2 累積和	7
2.1 一次元の累積和 (1)	7
2.2 一次元の累積和 (2)	10
2.3 二次元の累積和 (1)	10
2.4 二次元の累積和 (2)	10
2.5 チャレンジ問題	10

---

\*<https://atcoder.jp/contests/tessoku-book>

## 0 はじめに

この PDF は関数型言語 Haskell の習得のために、「競技プログラミングの鉄則」の問題に対して Haskell で記述したコード例をまとめた資料です。私のプログラミング力の向上を主目的とし、Haskell を学びたい人への手助けとなると良いと考えています。Haskell の I/O<sup>1</sup>は一筋縄でいかないので、アルゴリズム以前の問題で苦戦してしまうかもしれません。ですが、入出力に時間を取られるのは勿体無いため、そのような面でも Haskell 習得の支えになれば良いと願います。

## 1 アルゴリズムと計算量

### 1.1 導入問題

**問題.** 整数  $N$  が与えられるので、一辺の長さが  $N$  であるような正方形の面積を出力するプログラムを作成してください。

Listing 1: A01.hs

```
1 main :: IO ()
2 main = do
3     n <- readLn :: IO Int
4     putStrLn $ show (n * n)
```

**解説.** `readLn` の型は `Read a => IO a` です。ここで、`:: IO Int` と後ろに記述することで型変数  $a$  は `Int` 型であると宣言することができます。もちろん、書かなくても `(*)` という演算が後ろで行われていることから、型クラス `Num` に属することを推論してくれます。(書いた方が見やすいのではないのでしょうか?) `n <- readLn :: IO Int` を実行すると、 $n$  は通常の整数値になります。`show` の型は `Show a => a -> String` ですので、`n * n` の結果を `show` に渡すことで、整数値を文字列に変換することができます。最後に `String -> IO ()` という型をもつ `putStrLn` に渡せば完了です。

**問題.** 整数  $A$  と  $B$  が与えられるので、 $A + B$  の値を出力するプログラムを作成してください。ただし、制約は  $1 \leq A \leq 100$ 、 $1 \leq B \leq 100$  であるとします。

Listing 2: B01.hs

```
1 main :: IO ()
2 main = do
3     [a, b] <- map read . words <$> getLine
4     putStrLn $ show (a + b)
```

**解説.** まずは2つの空白区切りの整数の入力から説明します。`<$>`について、定義は以下の通りです。

```
(<$>) :: (Functor f) => (a -> b) -> f a -> f b
f <$> x = fmap f x
```

`<$>`の型に注目して下さい。`IO` は `Functor` のインスタンスですので、 $f$  を `IO` に置き換えると、`<$> :: (a -> b) -> IO a -> IO b` となります。また、

<sup>1</sup>I/O : Input(標準入力), Output(標準出力)

- `getLine :: IO String`
- `map read . words :: Read b => String -> [b]`

と定義されています。以上を合わせると

```
map read . words <$> getLine :: (Read b) => IO [b]
```

となることが分かります。IO [b] については、コンパイラが勝手に IO [Int] としてくれます。<sup>2</sup> 当然、`:: IO Int` と型を指定するのもいいことでしょう。最終的に、結果を Int 型のリストとして `[a, b]` に束縛することができます。(➡注 入力が2つであると分かっているので、`[a, b]` としています。) あとは計算結果を A01 と同様に出力すれば完了です。

## 1.2 全探索1

**問題.**  $N$  個の整数  $A_1, A_2, \dots, A_N$  の中に、整数  $X$  が含まれるかどうかを判定するプログラムを作成してください。

Listing 3: A02.hs

```
1 main :: IO ()
2 main = do
3     [n, x] <- map read . words <$> getLine :: IO [Int]
4     as <- map read . words <$> getLine :: IO [Int]
5     putStrLn $ if x `elem` as then "Yes" else "No"
```

**解説.** 3, 4 行目に関しては Listing 2 の解説で紹介しました。そろそろ気付いたかもしれませんが、Haskell では `for` を使った繰り返し処理を行わないので、リストの長さを示す `n` を受け取る必要がないことが多いです。そのため、3 行目は

```
[_, x] <- map read . words <$> getLine
```

としても良いです。そして、リスト `as` の中に `x` が含まれているかは、関数 `elem` で判定することができます。関数 `elem` は中置記法を用いることが多いです。

**問題.**  $A$  以上  $B$  以下の整数のうち、100 の約数であるものは存在しますか。答えを Yes か No で出力するプログラムを作成してください。

Listing 4: B02.hs

```
1 main :: IO ()
2 main = do
3     [a, b] <- map read . words <$> getLine :: IO [Int]
4     let x = [a..b]
5     putStrLn $ if 0 `elem` map (100 `mod`) x then "Yes" else "No"
```

**解説.** 4 行目の `[a..b]` は、`[a, a+1, a+2, ..., b-1, b]` を表します。このリストのすべての要素について、100 から割った余りを考えます。その余りのリストの中に 0 が含まれているか否かを判定します。

<sup>2</sup>Type Defaulting というものが使われています。詳しくはいつか書きます。(February 1, 2024)

### 1.3 全探索2

**問題.** 赤いカードが  $N$  枚あり、それぞれ整数  $P_1, P_2, \dots, P_N$  が書かれています。また、青いカードが  $N$  枚あり、それぞれ整数  $Q_1, Q_2, \dots, Q_N$  が書かれています。太郎くんは、赤いカードの中から1枚、青いカードの中から1枚、合計2枚のカードを選びます。選んだ2枚のカードに書かれた整数の合計が  $K$  となるようにする方法は存在しますか。答えを出力するプログラムを書いてください。

Listing 5: A03.hs

```
1 main :: IO ()
2 main = do
3     [_ , k] <- map read . words <$> getLine :: IO [Int]
4     ps <- map read . words <$> getLine :: IO [Int]
5     qs <- map read . words <$> getLine :: IO [Int]
6     putStrLn $ if k `elem` [p + q | p <- ps, q <- qs] then "Yes"
                    else "No"
```

**解説.** 6行目の

```
[p + q | p <- ps, q <- qs]
```

では、リスト `ps` とリスト `qs` から一つずつ要素を取り出し、それらを足し合わせて可能性のある答えをすべて計算して提示しています。これを**非決定性計算**とみなすことができます。このとき、

```
(+) <$> ps <*> qs
```

と書くことができます。これについては**アプリカティブ**という用語を聞くまでは保留で構いません。非決定性計算という言葉も同様です。(いずれ、後者の記法の方がきれいだと感じる日が来るはずです！私は内包表記も好きです。)

**問題.**  $N$  個の商品があり、商品  $i$  ( $1 \leq i \leq N$ ) の価格は  $A_i$  円です。異なる3つの商品を選び、合計価格をピッタリ 1000 円にする方法は存在しますか。答えを Yes か No で出力するプログラムを作成してください。ただし、制約は  $3 \leq N \leq 1000$  であるとします。

Listing 6: B03.hs

```
1 just :: Int -> Int -> [Int] -> Bool
2 just _ _ [] = False
3 just 1 num xs = num `elem` xs
4 just 2 num (x : xs) = any (\m -> m + x == num) xs || just 2 num
    xs
5 just k num as@(x : xs) = length as >= k && k >= 0
6     && (just (k - 1) (num - x) xs || just k num xs)
7
8 main :: IO ()
9 main = do
10     _ <- read <$> getLine :: IO Int
11     as <- map read . words <$> getLine :: IO [Int]
12     putStrLn $ if just 3 1000 as then "Yes" else "No"
```

**解説.** 本問も非決定性計算だから、Listing 5 と同様に

```
1000 `elem` [x + y + z | x <- as, y <- as, z <- as]
```

で可能性のある答えを列挙しよう！としては... 方針は正しいですが、その手法に誤りが生じています。これでは、同じ商品を選ぶ場合も含まれてしまうからです。

それでは、関数 `just` について説明します。`just k num xs` とは、リスト `xs` から、異なる `k` 個の要素を取り出して足し合わせたとき、その答えが `num` になることがあるか否かを示す式です。したがって、本文では入力から得たリストに `just 3 1000` を適用します。

① `just _ _ []`

リストが空なら、問答無用で `False` を返します。

② `just 1 num xs`

リストから 1 個だけ取り出して、`num` に等しいかを判定します。おなじみの関数 `elem` を使えば解決です。

③ `just 2 num xs`

まず、`any (\m -> m + x == num) xs` で、先頭要素とその他の要素を足して `num` になるかを判定しています。そこで、`True` が得られなかった場合、`just 2 num xs` で先頭要素を捨てて残りのリストで同じことを試します。以降、再帰的に判定します。

④ `just k num xs`

`k` 個取り出すので、`k` が負でないこと、リストの長さが `k` 以上であることを判定しておきます。(➡注 このような判定を書かずとも、AtCoder では正しい入力を与えられることが保証されているので問題ありません。しかし、一般的にはそうではないので、意図せぬ入力にも備えておく癖をつけましょう！とは言っても、AtCoder ではこれを書くのは手間ですので、やっぱり以降はサボります。)

あとは、③と同様に、まずは先頭要素とその他の要素について条件を満たす組合せが存在するかを判定します。このとき、`just (k - 1) (num - x) xs` に帰着されることを理解しましょう。

## 1.4 2進法

**問題.** 整数  $N$  が 10 進法表記で与えられます。  $N$  を 2 進法に変換した値を出力するプログラムを作成してください。

Listing 7: A04.hs

```
1 toBinary :: Int -> String
2 toBinary 1 = "1"
3 toBinary n = toBinary (n `div` 2) ++ if even n then "0" else "1"
4
5 assignZero :: Int -> String -> String
6 assignZero 0 str = str
7 assignZero x str = "0" ++ assignZero (x - 1) str
8
9 main :: IO ()
10 main = do
11     n <- readLn :: IO Int
12     let binary = toBinary n
13     putStrLn $ assignZero (10 - length binary) binary
```

**解説.** 関数 `toBinary` の動きを見てみましょう。

```
toBinary 12 -> toBinary 6 ++ "0" -> toBinary 3 ++ "00"
           -> toBinary 1 ++ "100" -> "1100"
```

どうでしょう？2進数を計算する方法は数 A で習いますね！

**問題.** 整数  $N$  (8 桁以内) が 2 進法表記で与えられます。  $N$  を 10 進法に変換した値を出力するプログラムを作成してください。

Listing 8: B04.hs

---

```
1 bin2list :: Int -> [Int] -> [Int]
2 bin2list 0 ys = ys
3 bin2list x ys = bin2list (x `div` 10) (x `mod` 10 : ys)
4
5 list2dec :: [Int] -> Int
6 list2dec [] = 0
7 list2dec (x : xs) = x * (2 ^ length xs) + list2dec xs
8
9 main :: IO ()
10 main = do
11     n <- readLn :: IO Int
12     print $ list2dec $ bin2list n []
```

---

**解説.** 与えられた入力 (例えば 101011) を

```
101011 -> [1,0,1,0,1,1] -> 2^5 + 2^3 + 2^1 + 2^0 -> 43
```

という流れで計算しています。入力を `Int` として受け取り、それをリスト構造に変換しているなら、最初からリスト構造である `String` で受け取るべきかもしれませんね。

おっと、関数 `print` が初登場しました。型は以下のようにになっています。 `putStrLn` との違いを見てみましょう。

```
print :: Show a => a -> IO ()
putStrLn :: String -> IO ()
```

## 1.5 チャレンジ問題

**問題.** 赤・青・白の 3 枚のカードがあります。太郎くんは、それぞれのカードに 1 以上  $N$  以下の整数を書かなければなりません。3 枚のカードの合計を  $K$  にするような書き方は何通りありますか。

Listing 9: A05.hs

---

```
1 main :: IO ()
2 main = do
3     [n, k] <- map read . words <$> getLine :: IO [Int]
4     print $ length [(a, b, c) | a <- [1..n], b <- [1..n], let c =
        k - a - b, 1 <= c, c <= n]
```

---

**解説.** リスト内包表記のおかげで解説いらず！1 以上  $n$  以下の範囲で、 $a + b + c = k$  となるように  $c$  を定めてあげます。出来上がったトリプルのリストの長さが答えになります。

## 2 累積和

### 2.1 一次元の累積和(1)

**問題.** ある遊園地では  $N$  日間にわたるイベントが開催され、 $i$  日目には  $A_i$  人が来場しました。以下の  $Q$  個の質問に答えるプログラムを作成してください。

- 質問 1:  $L_1$  日目から  $R_1$  までの来場者数は?
- $\vdots$
- 質問  $Q$ :  $L_Q$  日目から  $R_Q$  までの来場者数は?

Listing 10: A06.00.hs

```

1 import Control.Monad ( replicateM_ )
2
3 cumulativeSum :: [Int] -> [Int]
4 cumulativeSum = scanl (+) 0
5
6 solve :: [Int] -> [Int] -> Int
7 solve s [l, r] = (s !! r) - (s !! (l - 1))
8
9 main :: IO ()
10 main = do
11     [_, q] <- map read . words <$> getLine :: IO [Int]
12     as <- map read . words <$> getLine :: IO [Int]
13     let cumSum = cumulativeSum as
14     replicateM_ q $ do
15         lr <- map read . words <$> getLine :: IO [Int]
16         print $ solve cumSum lr

```

**解説.** 新しい関数がたくさん登場しました。一つずつ紹介します。

関数 `scanl` の定義は以下です。

```

scanl :: (b -> a -> b) -> b -> [a] -> [b]
scanl = scanlGo
  where
    scanlGo :: (b -> a -> b) -> b -> [a] -> [b]
    scanlGo f q ls = q : (case ls of
        []    -> []
        x:xs  -> scanlGo f (f q x) xs)

```

なぜ、`scanlGo` を挟んでいるのか!? よく分かっていませんが、実装自体は理解できそうですね! 動作を見てみましょう。

```
scanl (+) 0 [1,2,3,4,5] -> [0,1,3,6,10,15]
```

これがまさに累積和になります。引数のリストの長さを  $N$  とすると、関数 `cumulativeSum` は  $O(N)$  となります。上記の例で考えると、2 日目から 4 日目までの来場者数は、 $10 - 1 = 9$  (人) となります。3 日目から 5 日目までの来場者数は、 $15 - 3 = 12$  (人) となります。

リストの  $k$  番目の要素にアクセスするために関数 `(!!)` が使えます。したがって、質問に回答する関数 `solve` は

```
solve s [1, r] = (s !! r) - (s !! (1 - 1))
```

と定義できます。(「0 日目は累計 0 人」という無意味とも思える情報をリストの先頭に置いておくことで、リストのインデックスと日にちの数字が同じ数字になるので扱いやすくなっています。)これを、関数 `replicateM` を用いて  $q$  個の質問に適用します。ここでは、使用例だけ紹介します。

```
ghci> replicateM_ 3 (putStrLn "a")
a
a
a
```

これで完成！いえ、実行時間超過です。なぜでしょうか... オーダーに問題があるそうです。プログラムとしては正しい出力が得られますが、このプログラムは  $O(N^2 + Q)$  となっています。 $0 \leq N \leq 10^5$  ですので、最悪計算量が  $10^{10}$  となり制限時間超過となります。問題文では、 $O(N + Q)$  であることが望ましいと書かれていますので、それを目標にプログラムを変更しましょう。

関数 `cumulativeSum` は  $O(N)$  で、関数 `replicateM` は  $O(Q)$  です。したがって、関数 `(!!)` が  $O(1)$  であれば、非常に嬉しい！ということになります。しかしながら、関数 `(!!)` は  $O(N)$  であるようです。下記の定義からも確認できますね。

```
(!!) :: [a] -> Int -> a
xs      !! n | n < 0 = errorWithoutStackTrace
                  "Prelude.!!: negative index"
[]      !! _       = errorWithoutStackTrace
                  "Prelude.!!: index too large"
(x:_ )  !! 0       = x
(_:xs ) !! n       = xs !! (n-1)
```

では、 $O(1)$  でリストの要素にアクセスするにはどうしたらよいのでしょうか？ `Data.Array` モジュールを使いましょう。関数 `listArray` はインデックスの範囲とリストを引数に取り、インデックスの付いた配列を返してくれます。

```
listArray (0, 5) [0,1,3,6,10,15]
-> array (0, 5) [(0,0), (1,1), (2,3), (3,6), (4,10), (5,15)]
```

配列では、 $i$  番目の要素に  $O(1)$  でアクセスでき、アクセスする関数は `(!)` となっています。

これらを用いて変更したプログラムは Listing 11 のようになります。関数 `cumulativeSum` では、累積和を取ったのちに関数 `listArray` を適用しています。関数 `solve` では、関数 `(!!)` が関数 `(!)` に変わっただけですね！

Listing 11: A06\_01.hs

```
1 import Control.Monad ( replicateM_ )
2 import Data.Array ( Array, (!), listArray )
3
4 cumulativeSum :: Int -> Int -> [Int] -> Array Int Int
5 cumulativeSum l r as = listArray (l, r) $ scanl (+) 0 as
6
7 solve :: Array Int Int -> (Int, Int) -> Int
8 solve s (l, r) = (s ! r) - (s ! (1 - 1))
9
```



```

10 main :: IO ()
11 main = do
12     [n, q] <- map read . words <$> getLine :: IO [Int]
13     as <- map read . words <$> getLine :: IO [Int]
14     let cumSum = cumulativeSum 0 n as
15     replicateM_ q $ do
16         [l, r] <- map read . words <$> getLine :: IO [Int]
17         print $ solve cumSum (l, r)

```

**問題.** 太郎君はくじを  $N$  回引き、 $i$  回目の結果は  $A_i$  でした。 $A_i = 1$  のときアタリ、 $A_i = 0$  のときハズレを意味します。「 $L$  回目から  $R$  回目までの中では、アタリとハズレどちらが多いか？」という形式の質問が  $Q$  個与えられるので、それぞれの質問に答えるプログラムを作成してください。計算量は  $O(N + Q)$  であることが望ましいです。

### Listing 12: B06.hs

```

1 import Control.Monad ( replicateM_ )
2 import Data.Array ( Array, (!), listArray )
3
4 cumulativeSum :: (Int, Int) -> [Int] -> Array Int Int
5 cumulativeSum lr xs = listArray lr $ scanl (+) 0 xs
6
7 solve :: Array Int Int -> (Int, Int) -> Int
8 solve s (l, r) = (s ! r) - (s ! (l - 1))
9
10 main :: IO ()
11 main = do
12     n <- readLn :: IO Int
13     as <- map read . words <$> getLine :: IO [Int]
14     q <- readLn :: IO Int
15     let cumSum = cumulativeSum (0, n) as
16     replicateM_ q $ do
17         [l, r] <- map read . words <$> getLine :: IO [Int]
18         let t = solve cumSum (l, r)
19         putStrLn $ case (r - l + 1 - t) `compare` t of
20             GT -> "lose"
21             EQ -> "draw"
22             LT -> "win"

```

**解説.** 累積和を計算するところまで全問と全く同じです。 $t = \text{solve cumSum } (l, r)$  としていますが、この  $t$  は何を表しているのでしょうか。具体例で考えてみましょう。 $as$  が

```
[0, 1, 1, 0, 1, 0, 0]
```

のとき、 $\text{cumSum}$  は

```
array (0,7) [(0,0),(1,0),(2,1),(3,2),(4,2),(5,3),(6,3),(7,3)]
```

となります。リストの長さが 1 増えていますが、全問と同様に「0 回目はアタリは 0 個」という情報を付加しているためです。いや、その情報は必要ないんだ！という場合は、累積和の計算を `scanl1 (+) xs` としましょう。ここで、2 回目から 5 回目までについて考えてみます。2, 3, 5 回目でアタリが出ているので、アタリ 3 個・ハズレ 1 個となります。このアタリ 3 個がまさに  $t$  の値で、`solve cumSum (2, 5)` によって求まります。このとき、ハズレの個数は  $r - l + 1 - t$  と表現できます。あとは、この 2 つを比較して出力する文字を定めれば完了です。

## 2.2 一次元の累積和 (2)

**問題.** ある会社では  $D$  日間にわたってイベントが開催され、 $N$  人が出席します。参加者  $i$  は  $L_i$  日目から  $R_i$  日目まで出席する予定です。各日の出席者数を出力するプログラムを作ってください。

## 2.3 二次元の累積和 (1)

## 2.4 二次元の累積和 (2)

## 2.5 チャレンジ問題