



競技プログラミングの鉄則*

やさしい理系お兄ちゃん

June 19, 2024

Contents

1	アルゴリズムと計算量	2
1.1	導入問題	2
1.1.1	問題 A01 The First Problem	2
1.2	全探索 1	4
1.3	全探索 2	5
1.4	2進法	7
1.5	チャレンジ問題	8
2	累積和	8
2.1	一次元の累積和 (1)	8
2.2	一次元の累積和 (2)	12
2.3	二次元の累積和 (1)	13
2.4	二次元の累積和 (2)	15
2.5	チャレンジ問題	17
3	二分探索	18
3.1	配列の二分探索	18
3.2	答えで二分探索	19
3.3	しゃくとり法	21
3.4	半分全列挙	22
3.5	チャレンジ問題	24

*<https://atcoder.jp/contests/tessoku-book>

1 アルゴリズムと計算量

1.1 導入問題

1.1.1 問題 A01 The First Problem

問題. 整数 N が与えられるので、一辺の長さが N であるような正方形の面積を出力するプログラムを作成してください。

解説. 解答は以下のようになります。

```
1 main = print . (^2) =<< readLn
```

おっと、いきなり驚かせてしまいました。訳わからないですね。Haskell で競プロを行うにあたって、入出力をどうするかが最初の大きな壁となります。入出力はモナドという概念が使われており、そのモナドを理解することがそもそも難しいからです。ここで、上記のコードを以下のように書き換えてみます。

```
1 main = do
2     n <- readLn
3     print (n ^ 2)
```

何をやっているのかが見えてきましたね！`n <- readLn` で `n` に入力値を受け取っています。そして、答えの n^2 を `print (n ^ 2)` で出力します。このように、**do 記法** (do 構文) を用いることで命令型のプログラムに似た形で記述ができます。

ここから少し難しい話に入りますので、最初は簡単に読む程度で問題ありません。まずは GHC の対話モードを起動して `readLn` の型を問い合わせてみましょう。

```
ghci> :t readLn
readLn :: Read a => IO a
```

`a` を型クラス `Read` の**インスタンス**の型として、`readLn` の型は `IO a` となります。言い換えると、「`readLn` は `a` を生成する IO アクションである」となります。

```
n <- readLn
```

と書くことで、IO の中身の型 `a` の値を `n` に束縛することができます。すなわち、`n :: Read a => a` となります。ではここで、エディタの拡張機能に `n` の型を教えてもらいましょう。

```
modu n :: Integer
main : Defined at /Users/yasari/Desktop/atcoder/AtCoder-with-Haskell/tessoku-
main : book/a01/main.hs:5:5
      n <- readLn
      print (n ^ 2)
```

あれ、`n :: Integer` となっています。Read のインスタンスは `Int` や `Float`、`(Int, Char)` など様々な型があります。数ある候補の中からなぜ `Integer` が選ばれたのでしょうか？それは、Haskell の**型検査器 (type checker)** が賢いからです。後ろで `n` が `n ^ 2` という使われ方をしているため、それなら `n` の型は `Integer` だな！と**型推論**をしてくれます。いやいや、ちょっと待って。俺は `n` を `Int` 型として扱いたいんだ！という場合は

```
n <- readLn :: IO Int
```

と、型注釈をつけてあげましょう。初めての AC！嬉しいですね。

コンパイルエラー

```

app/Main.hs:4:1: warning: [-Wmissing-signatures]
  Top-level binding with no type signature: main :: IO ()
|
4 | main = do
|   ^^^^

app/Main.hs:5:10: warning: [-Wtype-defaults]
  • Defaulting the type variable 'a0' to type 'Integer' in the following constraints
    (Read a0) arising from a use of 'readLn' at app/Main.hs:5:10-15
    (Show a0) arising from a use of 'print' at app/Main.hs:6:5-9
    (Num a0) arising from a use of '^' at app/Main.hs:6:14
  • In a stmt of a 'do' block: n <- readLn
    In the expression:
      do n <- readLn
        print (n ^ 2)
    In an equation for 'main':
      main
        = do n <- readLn
          print (n ^ 2)
|
5 |   n <- readLn
|     ^^^^^^

app/Main.hs:6:14: warning: [-Wtype-defaults]
  • Defaulting the type variable 'b0' to type 'Integer' in the following constraints
    (Integral b0) arising from a use of '^' at app/Main.hs:6:14
    (Num b0) arising from the literal '2' at app/Main.hs:6:16
  • In the first argument of 'print', namely '(n ^ 2)'
    In a stmt of a 'do' block: print (n ^ 2)
    In the expression:
      do n <- readLn
        print (n ^ 2)
|
6 |   print (n ^ 2)
|     ^

```

解答例.

Listing 1: 問題 A01 The First Problem

```

1 main :: IO ()
2 main = do
3     n <- readLn
4     print (n * n)

```

問題. 整数 A と B が与えられるので、 $A + B$ の値を出力するプログラムを作成してください。ただし、制約は $1 \leq A \leq 100$ 、 $1 \leq B \leq 100$ であるとします。

Listing 2: B01.hs

```
1 main :: IO ()
2 main = do
3     [a, b] <- map read . words <$> getLine :: IO [Int]
4     print $ a + b
```

解説. まずは2つの空白区切りの整数の入力から説明します。初学者にとってはいきなり難しい話になります。関数 `getLine` で文字列を1行分読み取ることができます。`getLine` の型は `getLine :: IO String` となっています。うーん... 私たちは、中身の `String` の部分を整数値のリストに変換して、IO モナドから中身を取り出したいのです。どうすれば良いのでしょうか。具体例を用いて説明します。

例えば、入力が「3 2」のとき、`IO String` の中身の `String` は「3 2」という文字列になっています。この文字列に対して関数 `words` を適用すると、`["3", "2"]` というように空白区切りの文字列をリストに分解することができます。さらに、各要素に対して関数 `read` を適用することで、`[3, 2]` と整数値のリストにすることができます。では、実際に `IO String` という型を持つ値にどのようにして `words` や `read` を適用させることができるのでしょうか？

それを実現する関数が `<$>` となります。`<$>` を用いることで、IO モナドに包まれた値に対して関数を適用することができます。`<$>` について、定義は以下の通りです。

```
(<$>) :: (Functor f) => (a -> b) -> f a -> f b
f <$> x = fmap f x
```

`<$>` の型に注目して下さい。IO は `Functor` のインスタンスですので、`f` を IO に置き換えると、`<$> :: (a -> b) -> IO a -> IO b` となります。また、

- `getLine :: IO String`
- `map read . words :: Read b => String -> [b]`

と定義されています。以上を合わせると

```
map read . words <$> getLine :: (Read b) => IO [b]
```

となることが分かります。IO [b] については、`:: IO [Int]` と型注釈をつけてあげること、型を指定することができます。慣れるまで入力を受け取るときは、どのような型を扱っているかを意識するためにも型注釈をつけておきましょう。最終的に、`<->` を使うことで、IO モナドの中身である `Int` 型のリストを `[a, b]` に束縛することができます。(➡注 入力が2つであると分かっているので、`[a, b]` としています。) あとは計算結果を A01 と同様に出力すれば完了です。

1.2 全探索1

問題. N 個の整数 A_1, A_2, \dots, A_N の中に、整数 X が含まれるかどうかを判定するプログラムを作成してください。

Listing 3: A02.hs

```
1 main :: IO ()
2 main = do
3     [n, x] <- map read . words <$> getLine :: IO [Int]
4     as <- map read . words <$> getLine :: IO [Int]
5     putStrLn $ if x `elem` as then "Yes" else "No"
```

解説. 3, 4行目に関してはListing 2の解説で紹介しました。よく見たら n を使っていませんね。そのため、3行目は

```
[_, x] <- map read . words <$> getLine
```

としても良いです。そして、リスト as の中に要素 x が含まれているかは、関数 `elem` で判定することができます。関数 `elem` は中置記法を用いることが多いです。関数 `putStrLn` は `String` を受け取って `IO ()` を返します。なぜ関数 `print` を使わないのかは、使ってみるとわかるでしょう！

問題. A 以上 B 以下の整数のうち、100 の約数であるものは存在しますか。答えを Yes か No で出力するプログラムを作成してください。

Listing 4: B02.hs

```
1 main :: IO ()
2 main = do
3     [a, b] <- map read . words <$> getLine :: IO [Int]
4     let xs = [a..b]
5     putStrLn $ if 0 `elem` map (100 `mod`) xs then "Yes" else "No"
```

解説. 4行目の $[a..b]$ は、 $[a, a+1, a+2, \dots, b-1, b]$ を表します。このリストのすべての要素について、100 から割った余りを考えます。その余りのリストの中に 0 が含まれているか否かを判定します。

1.3 全探索2

問題. 赤いカードが N 枚あり、それぞれ整数 P_1, P_2, \dots, P_N が書かれています。また、青いカードが N 枚あり、それぞれ整数 Q_1, Q_2, \dots, Q_N が書かれています。太郎くんは、赤いカードの中から1枚、青いカードの中から1枚、合計2枚のカードを選びます。選んだ2枚のカードに書かれた整数の合計が K となるようにする方法は存在しますか。答えを出力するプログラムを書いてください。

Listing 5: A03.hs

```
1 main :: IO ()
2 main = do
3     [_, k] <- map read . words <$> getLine :: IO [Int]
4     ps <- map read . words <$> getLine :: IO [Int]
5     qs <- map read . words <$> getLine :: IO [Int]
6     putStrLn $ if k `elem` [p + q | p <- ps, q <- qs] then "Yes"
                    else "No"
```

解説. 6 行目の

```
[p + q | p <- ps, q <- qs]
```

では、リスト `ps` とリスト `qs` から一つずつ要素を取り出し、それらを足し合わせて可能性のある答えをすべて計算して提示しています。とりあえず、足した結果をすべて用意しておくということです。これは**非決定性計算**と呼ばれます。このとき、

```
(+) <$> ps <*> qs
```

と書くことができます。これについては**アプリカティブ**という用語を聞くまでは保留で構いません。非決定性計算という言葉も同様です。いずれ、後者の記法の方がきれいだと感じる日が来るはずです！私はリストの内包表記も好きです。

問題. N 個の商品があり、商品 i ($1 \leq i \leq N$) の価格は A_i 円です。異なる 3 つの商品を選び、合計価格をピッタリ 1000 円にする方法は存在しますか。答えを Yes か No で出力するプログラムを作成してください。ただし、制約は $3 \leq N \leq 1000$ であるとします。

Listing 6: B03.hs

```
1 just :: Int -> Int -> [Int] -> Bool
2 just _ _ [] = False
3 just 1 num xs = num `elem` xs
4 just k num (x : xs) = just (k - 1) (num - x) xs || just k num xs
5
6 main :: IO ()
7 main = do
8   _ <- read <$> getLine :: IO Int
9   as <- map read . words <$> getLine :: IO [Int]
10  putStrLn $ if just 3 1000 as then "Yes" else "No"
```

解説. 本問も非決定性計算だから、Listing 5 と同様に

```
1000 `elem` [x + y + z | x <- as, y <- as, z <- as]
```

で可能性のある答えを列挙しよう！としては... 方針は正しいですが、その手法に誤りが生じています。これでは、同じ商品を選ぶ場合も含まれてしまうからです。

それでは、関数 `just` について説明します。`just k num xs` とは、リスト `xs` から、異なる k 個の要素を取り出して足し合わせたとき、その答えが `num` になることがあるか否かを示す式です。したがって、本文では入力から得たリストに `just 3 1000` を適用します。

① `just _ _ []`

リストが空なら、問答無用で `False` を返します。

② `just 1 num xs`

リストから 1 個だけ取り出して、`num` に等しいかを判定します。おなじみの関数 `elem` を使えば解決です。

③ `just 2 num xs`

まず、`any (\m -> m + x == num) xs` で、先頭要素とその他の要素を足して `num` になるかを判定しています。そこで、`True` が得られなかった場合、`just 2 num xs` で先頭要素を捨てて残りのリストで同じことを試します。以降、再帰的に判定します。

④ `just k num xs`

`k` 個取り出すので、`k` が負でないこと、リストの長さが `k` 以上であることを判定しておきます。(➡注 このような判定を書かずとも、AtCoder では正しい入力を与えられることが保証されているので問題ありません。しかし、一般的にはそうではないので、意図せぬ入力にも備えておく癖をつけましょう！とは言っても、AtCoder ではこれを書くのは手間ですので、やっぱり以降はサボります。)

あとは、③と同様に、まずは先頭要素とその他の要素について条件を満たす組合せが存在するかを判定します。このとき、`just (k - 1) (num - x) xs` に帰着されることを理解しましょう。

1.4 2進法

問題. 整数 N が 10 進法表記で与えられます。 N を 2 進法に変換した値を出力するプログラムを作成してください。

Listing 7: A04.hs

```
1 toBinary :: Int -> String
2 toBinary 1 = "1"
3 toBinary n = toBinary (n `div` 2) ++ if even n then "0" else "1"
4
5 assignZero :: Int -> String -> String
6 assignZero 0 str = str
7 assignZero x str = "0" ++ assignZero (x - 1) str
8
9 main :: IO ()
10 main = do
11     n <- readLn :: IO Int
12     let binary = toBinary n
13     putStrLn $ assignZero (10 - length binary) binary
```

解説. 関数 `toBinary` の動きを見てみましょう。

```
toBinary 12 -> toBinary 6 ++ "0" -> toBinary 3 ++ "00"
              -> toBinary 1 ++ "100" -> "1100"
```

どうでしょう？2進数を計算する方法は数 A で習いますね！

問題. 整数 N (8 桁以内) が 2 進法表記で与えられます。 N を 10 進法に変換した値を出力するプログラムを作成してください。

Listing 8: B04.hs

```
1 bin2list :: Int -> [Int] -> [Int]
2 bin2list 0 ys = ys
3 bin2list x ys = bin2list (x `div` 10) (x `mod` 10 : ys)
4
5 list2dec :: [Int] -> Int
6 list2dec [] = 0
7 list2dec (x : xs) = x * (2 ^ length xs) + list2dec xs
```



```

8
9 main :: IO ()
10 main = do
11     n <- readLn :: IO Int
12     print $ list2dec $ bin2list n []

```

解説. 与えられた入力（例えば101011）を

101011 -> [1,0,1,0,1,1] -> $2^5 + 2^3 + 2^1 + 2^0$ -> 43

という流れで計算しています。入力をIntとして受け取り、それをリスト構造に変換しているなら、最初からリスト構造であるStringで受け取るべきかもしれませんね。

おっと、関数printが初登場しました。型は以下のようになっています。putStrLnとの違いを見てみましょう。

```

print :: Show a => a -> IO ()
putStrLn :: String -> IO ()

```

1.5 チャレンジ問題

問題. 赤・青・白の3枚のカードがあります。太郎くんは、それぞれのカードに1以上 N 以下の整数を書かなければなりません。3枚のカードの合計を K にするような書き方は何通りありますか。

Listing 9: A05.hs

```

1 main :: IO ()
2 main = do
3     [n, k] <- map read . words <$> getLine :: IO [Int]
4     print $ length [(a, b, c) | a <- [1..n], b <- [1..n], let c =
        k - a - b, 1 <= c, c <= n]

```

解説. リスト内包表記のおかげで解説いらず！1以上 n 以下の範囲で、 $a + b + c = k$ となるように c を定めてあげます。出来上がったトリプルのリストの長さが答えになります。

2 累積和

2.1 一次元の累積和(1)

問題. ある遊園地では N 日間にわたるイベントが開催され、 i 日目には A_i 人が来場しました。以下の Q 個の質問に答えるプログラムを作成してください。

- 質問1: L_1 日目から R_1 までの来場者数は？
- ⋮
- 質問 Q : L_Q 日目から R_Q までの来場者数は？

Listing 10: A06.00.hs

```

1 import Control.Monad ( replicateM_ )
2
3 cumulativeSum :: [Int] -> [Int]
4 cumulativeSum = scanl (+) 0
5
6 solve :: [Int] -> [Int] -> Int
7 solve s [l, r] = (s !! r) - (s !! (l - 1))
8
9 main :: IO ()
10 main = do
11     [_, q] <- map read . words <$> getLine :: IO [Int]
12     as <- map read . words <$> getLine :: IO [Int]
13     let cumSum = cumulativeSum as
14     replicateM_ q $ do
15         lr <- map read . words <$> getLine :: IO [Int]
16         print $ solve cumSum lr

```

解説. 新しい関数がたくさん登場しました。一つずつ紹介します。
関数 `scanl` の定義は以下です。

```

scanl :: (b -> a -> b) -> b -> [a] -> [b]
scanl = scanlGo
  where
    scanlGo :: (b -> a -> b) -> b -> [a] -> [b]
    scanlGo f q ls = q : (case ls of
        []    -> []
        x:xs -> scanlGo f (f q x) xs)

```

なぜ、`scanlGo` を挟んでいるのか!? よく分かっていませんが、実装自体は理解できそうですね! 動作を見てみましょう。

```
scanl (+) 0 [1,2,3,4,5] -> [0,1,3,6,10,15]
```

これがまさに累積和になります。引数のリストの長さを N とすると、関数 `cumulativeSum` は $O(N)$ となります。上記の例で考えると、2 日目から 4 日目までの来場者数は、 $10 - 1 = 9$ (人) となります。3 日目から 5 日目までの来場者数は、 $15 - 3 = 12$ (人) となります。

リストの k 番目の要素にアクセスするために関数 `(!!)` が使えます。したがって、質問に回答する関数 `solve` は

```
solve s [l, r] = (s !! r) - (s !! (l - 1))
```

と定義できます。(「0 日目は累計 0 人」という無意味とも思える情報をリストの先頭に置いておくことで、リストのインデックスと日にちの数字が同じ数字になるので扱いやすくなっています。) これを、関数 `replicateM_` を用いて q 個の質問に適用します。ここでは、使用例だけ紹介します。

```

ghci> replicateM_ 3 (putStrLn "a")
a
a
a

```

これで完成！いえ、実行時間超過です。なぜでしょうか... オーダーに問題があるそうです。プログラムとしては正しい出力が得られますが、このプログラムは $O(N^2 + Q)$ となっています。 $0 \leq N \leq 10^5$ ですので、最悪計算量が 10^{10} となり制限時間超過となります。問題文では、 $O(N + Q)$ であることが望ましいと書かれていますので、それを目標にプログラムを変更しましょう。

関数 `cumulativeSum` は $O(N)$ で、関数 `replicateM` は $O(Q)$ です。したがって、関数 `(!!)` が $O(1)$ であれば、非常に嬉しい！ということになります。しかしながら、関数 `(!!)` は $O(N)$ であるようです。下記の定義からも確認できますね。

```
(!!) :: [a] -> Int -> a
xs    !! n | n < 0 =  errorWithoutStackTrace
                    "Prelude.!!: negative index"
[]    !! _       =  errorWithoutStackTrace
                    "Prelude.!!: index too large"

(x:_) !! 0       =  x
(_:xs) !! n      =  xs !! (n-1)
```

では、 $O(1)$ でリストの要素にアクセスするにはどうしたらよいのでしょうか？ `Data.Array` モジュールを使いましょう。関数 `listArray` はインデックスの範囲とリストを引数に取り、インデックスの付いた配列を返してくれます。

```
listArray (0, 5) [0,1,3,6,10,15]
-> array (0, 5) [(0,0), (1,1), (2,3), (3,6), (4,10), (5,15)]
```

配列では、 i 番目の要素に $O(1)$ でアクセスでき、アクセスする関数は `(!)` となっています。

これらを用いて変更したプログラムは Listing 11 のようになります。関数 `cumulativeSum` では、累積和を取ったのちに関数 `listArray` を適用しています。関数 `solve` では、関数 `(!!)` が関数 `(!)` に変わっただけです！

Listing 11: A06_01.hs

```
1 import Control.Monad ( replicateM )
2 import Data.Array ( Array, (!), listArray )
3
4 cumulativeSum :: Int -> Int -> [Int] -> Array Int Int
5 cumulativeSum l r as = listArray (l, r) $ scanl (+) 0 as
6
7 solve :: Array Int Int -> (Int, Int) -> Int
8 solve s (l, r) = (s ! r) - (s ! (l - 1))
9
10 main :: IO ()
11 main = do
12     [n, q] <- map read . words <$> getLine
13     as <- map read . words <$> getLine
14     lrs <- map ((\[l, r] -> (l, r)) . map read . words) <$>
15         replicateM q getLine
16     mapM_ (print . solve (cumulativeSum 0 n as)) lrs
```

問題. 太郎君はくじを N 回引き、 i 回目の結果は A_i でした。 $A_i = 1$ のときアタリ、 $A_i = 0$ のときハズレを意味します。「 L 回目から R 回目までの中では、アタリとハズレどちらが多いか？」という形式の質問が Q 個与えられるので、それぞれの質問に答えるプログラムを作成してください。計算量は $O(N + Q)$ であることが望ましいです。

Listing 12: B06.hs

```

1 import Control.Monad ( replicateM_ )
2 import Data.Array ( Array, (!), listArray )
3
4 cumulativeSum :: (Int, Int) -> [Int] -> Array Int Int
5 cumulativeSum lr xs = listArray lr $ scanl (+) 0 xs
6
7 solve :: Array Int Int -> (Int, Int) -> Int
8 solve s (l, r) = (s ! r) - (s ! (l - 1))
9
10 main :: IO ()
11 main = do
12     n <- readLn :: IO Int
13     as <- map read . words <$> getLine :: IO [Int]
14     q <- readLn :: IO Int
15     let cumSum = cumulativeSum (0, n) as
16     replicateM_ q $ do
17         [l, r] <- map read . words <$> getLine :: IO [Int]
18         let t = solve cumSum (l, r)
19         putStrLn $ case (r - l + 1 - t) `compare` t of
20             GT -> "lose"
21             EQ -> "draw"
22             LT -> "win"

```

解説. 累積和を計算するところまで全問と全く同じです。 $t = \text{solve cumSum } (l, r)$ していますが、この t は何を表しているのでしょうか。具体例で考えてみましょう。as が

`[0, 1, 1, 0, 1, 0, 0]`

のとき、cumSum は

`array (0,7) [(0,0),(1,0),(2,1),(3,2),(4,2),(5,3),(6,3),(7,3)]`

となります。リストの長さが 1 増えていますが、前問と同様に「0 回目はアタリ 0 個」という情報を付加しているためです。いや、その情報は必要ないんだ！という場合は、累積和の計算を `scanl1 (+) xs` としましょう。ここで、2 回目から 5 回目までについて考えてみます。2, 3, 5 回目でアタリが出ているので、アタリ 3 個・ハズレ 1 個となります。このアタリ 3 個がまさに t の値で、`solve cumSum (2, 5)` によって求まります。このとき、ハズレの個数は $r - l + 1 - t$ と表現できます。あとは、この 2 つを比較して出力する文字を定めれば完了です。

2.2 一次元の累積和 (2)

問題. ある会社では D 日間にわたってイベントが開催され、 N 人が出席します。参加者 i は L_i 日目から R_i 日目まで出席する予定です。各日の出席者数を出力するプログラムを作ってください。

Listing 13: A07.hs

```
1 import Control.Monad ( replicateM )
2 import Data.Array ( accumArray, elems, Array )
3
4 add2PreviousDay :: [Int] -> [(Int, Int)]
5 add2PreviousDay [l, r] = [(l, 1), (r + 1, -1)]
6 add2PreviousDay _ = []
7
8 cumulativeSum :: Array Int Int -> [Int]
9 cumulativeSum xs = scanl (+) 0 $ elems xs
10
11 main :: IO ()
12 main = do
13     d <- readLn
14     n <- readLn
15     lrs <- concatMap (add2PreviousDay . fmap read . words) <$>
16         replicateM n getLine
17     let accArray = accumArray (+) 0 (1, d + 1) lrs
18     mapM_ print $ tail $ init $ cumulativeSum accArray
```

解説. 例えば、イベントが5日間開催されたとして、やさ理くんは2日目から4日目まで参加したとしましょう。このとき、リスト $[0, 1, 0, 0, -1]$ の累積和をとると、 $[0, 0, 1, 1, 1, 0]$ となり、やさ理くんが出席した日のみ1となっていることが分かります。もちろん、ここでも「0 日目は出席者0人」という情報を付加しています。

以上のように、 L 日目から R 日目まで参加した人に関して、 $[(L, 1), (R+1, -1)]$ というリストを用意しておき、上手く累積和をとることができれば良いわけです。その上手くやる方法を説明していきます。

問題. あるコンビニは時刻 0 に開店し、時刻 T に閉店します。このコンビニでは N 人の従業員が働いており、 i 人目の従業員の出勤時刻は L_i 、退勤時刻は R_i です (L_i, R_i は整数)。 $t = 0, 1, \dots, T-1$ について、時刻 $t+0.5$ には何人の従業員が働いているかを出力するプログラムを作成してください。計算量は $O(N+T)$ であることが望ましいです。

Listing 14: B07.hs

```
1 import Control.Monad ( replicateM )
2 import Data.Array ( accumArray, elems, Array )
3
4 attendAndLeave :: [Int] -> [(Int, Int)]
5 attendAndLeave [l, r] = [(l, 1), (r, -1)]
6 attendAndLeave _ = []
7
8 cumulativeSum :: Array Int Int -> [Int]
```

```

9 cumulativeSum xs = scanl (+) 0 $ elems xs
10
11 main :: IO ()
12 main = do
13     t <- readLn
14     n <- readLn
15     lrs <- concatMap (attendAndLeave . fmap read . words) <$>
        replicateM n getLine
16     let accArray = accumArray (+) 0 (0, t) lrs
17     mapM_ print $ tail $ init $ cumulativeSum accArray

```

解説.

2.3 二次元の累積和(1)

問題. $H \times W$ のマス目があります。上から i 行目、左から j 列目にあるマス (i, j) には、整数 $X_{i,j}$ が書かれています。以下の Q 個の質問に答えるプログラムを作成してください。

- 質問 1: 左上 (A_1, B_1) 右下 (C_1, D_1) の長方形領域に書かれた整数の総和は?
- 質問 2: 左上 (A_2, B_2) 右下 (C_2, D_2) の長方形領域に書かれた整数の総和は?
- \vdots
- 質問 Q : 左上 (A_Q, B_Q) 右下 (C_Q, D_Q) の長方形領域に書かれた整数の総和は?

Listing 15: A08.hs

```

1 import Control.Monad ( replicateM )
2 import Data.Array ( Array, (!), listArray )
3
4 twoDimensionalSum :: Int -> Int -> [[Int]] -> Array (Int, Int)
   Int
5 twoDimensionalSum h w xs =
6     listArray ((0, 0), (h, w))
7     $ concat
8     $ scanl (zipWith (+)) (replicate (w + 1) 0)
9     $ map (scanl (+) 0) xs
10
11 solve :: Array (Int, Int) Int -> [[Int]] -> [Int]
12 solve twoDimSum = map (\[x, y, z, w] ->
13     twoDimSum ! (x - 1, y - 1) + twoDimSum ! (z, w)
14     - twoDimSum ! (z, y - 1) - twoDimSum ! (x - 1, w))
15
16 main :: IO ()
17 main = do
18     [h, w] <- map read . words <$> getLine
19     xs <- map (map read . words) <$> replicateM h getLine
20     q <- readLn

```

```

21     qs <- map (map read . words) <$> replicateM q getLine
22     mapM_ print $ solve (twoDimensionalSum h w xs) qs

```

解説.

問題. 二次元平面上に N 個の点があります。 i 個目の点の座標は (X_i, Y_i) です。「 x 座標が a 以上 c 以下であり、 y 座標が b 以上 d 以下であるような点は何個あるか？」という形式の質問が Q 個与えられるので、それぞれの質問に答えるプログラムを実装してください。 $N \leq 100000$, $Q \leq 100000$, $1 \leq X_i, Y_i \leq 1500$ を満たすケースで、1 秒以内に実行が終わることが望ましいです。なお、入力される値はすべて整数です。

Listing 16: B08.hs

```

1 import Control.Monad ( replicateM )
2 import Data.Array ( Array, (!), listArray, accumArray, bounds )
3
4 twoDimensionalSum :: Int -> Int -> [[Int]] -> Array (Int, Int)
   Int
5 twoDimensionalSum h w xs =
6     listArray ((0, 0), (h, w))
7     $ concat
8     $ scanl1 (zipWith (+))
9     $ map (scanl1 (+)) xs
10
11 arrayToLists :: Array (Int, Int) Int -> [[Int]]
12 arrayToLists arr = [[arr ! (i, j) | j <- [0..jMax]] | i <- [0..
   iMax]]
13     where rightBound = snd (bounds arr)
14           iMax = fst rightBound
15           jMax = snd rightBound
16
17 solve :: Array (Int, Int) Int -> [(Int, Int, Int, Int)] -> [Int]
18 solve twoDimSum = map (\(a, b, c, d) ->
19     twoDimSum ! (a - 1, b - 1) + twoDimSum ! (c, d)
20     - twoDimSum ! (c, b - 1) - twoDimSum ! (a - 1, d))
21
22 main :: IO ()
23 main = do
24     n <- readLn
25     xys <- map ((\[x, y] -> ((x, y), 1)) . map read . words) <$>
       replicateM n getLine
26     q <- readLn
27     qs <- map ((\[a, b, c, d] -> (a, b, c, d)) . map read . words
       ) <$> replicateM q getLine
28     let h = maximum $ map (\((x, _), _) -> x) xys
29         w = maximum $ map (\((_, y), _) -> y) xys
30     mapM_ print $ solve (twoDimensionalSum h w $ arrayToLists $
       accumArray (+) 0 ((0, 0), (h, w)) xys) qs

```

解説.

2.4 二次元の累積和 (2)

問題. ALGO 王国は $H \times W$ のマス目で表されます。最初は、どのマスにも雪が積もっていませんが、これから N 日間にわたって雪が降り続けます。上から i 行目・左から j 列目のマスを (i, j) とするとき、 i 日目には「マス (A_i, B_i) を左上とし、マス (C_i, D_i) を右下とする長方形領域」の積雪が 1 cm だけ増加することが予想されています。最終的な各マスの積雪を出力するプログラムを作成してください。

Listing 17: A09.hs

```

1 import Control.Monad ( replicateM )
2 import Data.Array ( Array, (!), listArray, accumArray, elems,
    bounds )
3 import qualified Data.ByteString.Char8 as BS
4 import Data.ByteString ( ByteString )
5 import Data.Maybe ( fromJust )
6 import Data.List.Split ( chunksOf )
7
8 readInt :: ByteString -> Int
9 readInt = fst . fromJust . BS.readInt
10
11 showInt :: Int -> ByteString
12 showInt = BS.pack . show
13
14 getPairs :: [Int] -> [((Int, Int), Int)]
15 getPairs [x,y,z,w] = [((x, y), 1), ((z + 1, w + 1), 1), ((x, w
    + 1), -1), ((z + 1, y), -1)]
16 getPairs _ = []
17
18 twoDimensionalSum :: Array (Int, Int) Int -> Array (Int, Int) Int
19 twoDimensionalSum arr = listArray bounds_
20     $ concat
21     $ scanl1 (zipWith (+))
22     $ map (scanl1 (+)) lists
23   where bounds_ = bounds arr
24         lists = chunksOf ((snd . snd) bounds_) $ elems arr
25
26 printArray :: Array (Int, Int) Int -> IO ()
27 printArray arr = do
28   let ((minX, minY), (maxX, maxY)) = bounds arr
29   mapM_ (\i -> BS.putStrLn $ BS.unwords [showInt (arr ! (i,j)) |
    j <- [minY .. maxY - 1]]) [minX .. maxX - 1]
30
31 main :: IO ()
32 main = do
33   [h, w, n] <- map readInt . BS.words <$> BS.getLine
34   xys <- concatMap (getPairs . map readInt . BS.words) <$>
    replicateM n BS.getLine

```



```

35     printArray $ twoDimensionalSum $ accumArray (+) 0 ((1,1), (h +
        1, w + 1)) xys

```

解説.

問題. 二次元平面上に N 枚の紙があります。それぞれの紙は、各辺が x 軸または y 軸に平行であるような長方形となっています。また、 i 枚目の紙の左下座標は (A_i, B_i) であり、右上座標は (C_i, D_i) です。1 枚以上の紙が置かれている部分の面積を求めてください。
 $N \leq 100000$, $0 \leq A_i < C_i \leq 1500$, $0 \leq B_i < D_i \leq 1500$ を満たすケースで、1 秒以内に実行が終わることが望ましいです。なお、入力される値はすべて整数です。

Listing 18: B09.hs

```

1  import Control.Monad ( replicateM )
2  import Data.Array ( Array, listArray, accumArray, elems, bounds )
3  import qualified Data.ByteString.Char8 as BS
4  import Data.ByteString ( ByteString )
5  import Data.Maybe ( fromJust )
6  import Data.List.Split ( chunksOf )
7
8  readInt :: ByteString -> Int
9  readInt = fst . fromJust . BS.readInt
10
11 showInt :: Int -> ByteString
12 showInt = BS.pack . show
13
14 getPairs :: [Int] -> [((Int, Int), Int)]
15 getPairs [x,y,z,w] = [((x, y), 1), ((z, w), 1), ((x, w), -1),
    ((z, y), -1)]
16 getPairs _ = []
17
18 twoDimensionalSum :: Array (Int, Int) Int -> Array (Int, Int) Int
19 twoDimensionalSum arr = listArray bounds_
20     $ concat
21     $ scanl1 (zipWith (+))
22     $ map (scanl1 (+)) lists
23     where bounds_ = bounds arr
24           lists = chunksOf (1 + (snd . snd) bounds_) $ elems arr
25
26 area :: Array (Int, Int) Int -> Int
27 area arr = length . concatMap (filter (/= 0)) $ lists
28     where bounds_ = bounds arr
29           lists = chunksOf ((snd . snd) bounds_) $ elems arr
30
31 main :: IO ()
32 main = do
33     n <- readLn
34     xys <- concatMap (getPairs . map readInt . BS.words) <$>
        replicateM n BS.getLine

```

```

35     let h = maximum $ map (\(x, _), _) -> x) xys
36     w = maximum $ map (\(_, y), _) -> y) xys
37     arr = accumArray (+) 0 ((0,0), (h, w)) xys
38     print . area . twoDimensionalSum $ arr

```

解説.

2.5 チャレンジ問題

問題. あるリゾートホテルには、1号室から N 号室までの N 個の部屋があります。 i 号室は A_i 人部屋です。このホテルでは D 日間にわたって工事が行われることになっており、 d 日目は L_d 号室から R_d 号室までの範囲を使うことができません。 $d = 1, 2, \dots, D$ について、 d 日目に使える中で最も大きい部屋は何人部屋であるか、出力するプログラムを作成してください。

Listing 19: A10.hs

```

1 import Control.Monad ( replicateM )
2 import qualified Data.ByteString.Char8 as BS
3 import Data.ByteString ( ByteString )
4 import Data.Maybe ( fromJust )
5 import Data.Array ( Array, listArray, (!) )
6
7 cumulativeMax :: (Int, Int) -> [Int] -> Array Int Int
8 cumulativeMax bounds_ as = listArray bounds_ $ scanl1 max as
9
10 cumulativeMaxFromRight :: (Int, Int) -> [Int] -> Array Int Int
11 cumulativeMaxFromRight bounds_ xs = listArray bounds_ $ scanr max (
    last xs) (init xs)
12
13 solve :: Array Int Int -> Array Int Int -> [Int] -> Int
14 solve cumMaxL cumMaxR [l, r] = max (cumMaxL ! (l - 1)) (cumMaxR !
    (r + 1))
15 solve _ _ _ = -1
16
17 main :: IO ()
18 main = do
19     [n] <- map readInt . BS.words <$> BS.getLine
20     as <- map readInt . BS.words <$> BS.getLine
21     [d] <- map readInt . BS.words <$> BS.getLine
22     lrs <- map (map readInt . BS.words) <$> replicateM d BS.getLine
23     let bounds_ = (1, n)
24         ps = cumulativeMax bounds_ as
25         qs = cumulativeMaxFromRight bounds_ as
26     mapM_ (print . solve ps qs) lrs
27
28 readInt :: ByteString -> Int
29 readInt = fst . fromJust . BS.readInt
30

```

```

31 showInt :: Int -> ByteString
32 showInt = BS.pack . show

```

解説.

3 二分探索

3.1 配列の二分探索

問題. 小さい順に並べられている、要素数 N の配列 $A = [A_1, A_2, \dots, A_N]$ があります。要素 X は配列 A の何番目に存在するかを出力してください。なお、この問題は単純な全探索（→ 1.2 節）でも解けますが、ここでは二分探索法を使って実装してください。

Listing 20: A11.hs

```

1 import qualified Data.ByteString.Char8 as BS
2 import Data.ByteString ( ByteString )
3 import Data.Maybe ( fromJust )
4 import Data.Array ( Array, listArray, (!) )
5
6 readInt :: ByteString -> Int
7 readInt = fst . fromJust . BS.readInt
8
9 showInt :: Int -> ByteString
10 showInt = BS.pack . show
11
12 binarySearch :: Int -> Int -> Int -> Array Int Int -> Int
13 binarySearch x min max as = if x == num then mid else binarySearch
    x newMin newMax as
14     where mid = (max + min) `div` 2
15           num = as ! mid
16           newMin = if x < num then min else mid + 1
17           newMax = if x < num then mid - 1 else max
18
19 main :: IO ()
20 main = do
21     [n, x] <- map readInt . BS.words <$> BS.getLine
22     as <- map readInt . BS.words <$> BS.getLine
23     let arr = listArray (1, n) as
24     print $ binarySearch x 1 n arr

```

解説.

問題. 小さい順に並んでいるとは限らない、要素数 N の配列 $A = [A_1, A_2, \dots, A_N]$ が与えられます。それについて、以下の Q 個の質問に答えるプログラムを作成してください。

- 1 個目の質問：配列 A には X_1 より小さい要素が何個あるか？

- 2 個目の質問：配列 A には X_2 より小さい要素が何個あるか？

⋮

- Q 個目の質問：配列 A には X_Q より小さい要素が何個あるか？

$N, Q \leq 100000$ を満たすケースで、1 秒以内に実行が終わることが望ましいです。

Listing 21: B11.hs

```
1 import qualified Data.ByteString.Char8 as BS
2 import Data.ByteString ( ByteString )
3 import Data.Maybe ( fromJust )
4 import Data.Array ( Array, listArray, (!) )
5 import Data.List ( sort )
6 import Control.Monad ( replicateM )
7
8 readInt :: ByteString -> Int
9 readInt = fst . fromJust . BS.readInt
10
11 showInt :: Int -> ByteString
12 showInt = BS.pack . show
13
14 lowerBound :: Int -> Int -> Array Int Int -> Int -> Int
15 lowerBound l r arr x
16   | x < arr ! l = l - 1
17   | arr ! r < x = r
18   | otherwise = if x <= arr ! mid then lowerBound l (mid - 1)
19                 arr x else lowerBound (mid + 1) r arr x
20                 where mid = (l + r) `div` 2
21
22 main :: IO ()
23 main = do
24   n <- readLn
25   as <- map readInt . BS.words <$> BS.getLine
26   q <- readLn
27   xs <- concatMap (map readInt . BS.words) <$> replicateM q BS.
28     getLine
29   let arr = listArray (1, n) (sort as)
30   mapM_ (BS.putStrLn . showInt . lowerBound 1 n arr) xs
```

解説.

3.2 答えで二分探索

問題. N 台のプリンターがあり、1 から N までの番号が付けられています。プリンター i は A_i 秒ごとにチラシを 1 枚印刷します。すなわち、スイッチを入れてから A_i 秒後、 $2A_i$ 秒後、 $3A_i$ 秒後 … に印刷します。すべてのプリンターのスイッチを同時に入れたとき、 K 枚目のチラシが印刷されるのは何秒後でしょうか。

Listing 22: A12.hs

```

1 import qualified Data.ByteString.Char8 as BS
2 import Data.ByteString ( ByteString )
3 import Data.Maybe ( fromJust )
4 import Data.Array ( Array, listArray, (!) )
5
6 readInteger :: ByteString -> Integer
7 readInteger = fst . fromJust . BS.readInteger
8
9 check :: Array Integer Integer -> Integer -> Integer -> Integer->
    Bool
10 check arr n k mid = sum_ arr mid n >= k
11     where
12         sum_ :: Array Integer Integer -> Integer -> Integer ->
            Integer
13         sum_ arr_ mid_ 1 = mid_ `div` (arr_ ! 1)
14         sum_ arr_ mid_ i = mid_ `div` (arr_ ! i) + sum_ arr_ mid_ (
            i - 1)
15
16 binarySearch :: Integer -> Integer -> Array Integer Integer ->
    Integer -> Integer -> Integer
17 binarySearch l r arr n k
18     | l == r = 1
19     | check arr n k mid = binarySearch l mid arr n k
20     | otherwise = binarySearch (mid + 1) r arr n k
21     where
22         mid = (l + r) `div` 2
23
24 main :: IO ()
25 main = do
26     [n, k] <- map readInteger . BS.words <$> BS.getLine
27     as <- map readInteger . BS.words <$> BS.getLine
28     let arr = listArray (1, n) as
29     let left = 1
30     let right = 1000000000
31     print $ binarySearch left right arr n k

```

解説.

問題. 正の整数 N が与えられます。 $x^3 + x = N$ を満たす正の実数 x を出力してください。ただし、絶対誤差が 0.001 以下であれば正解とします。

Listing 23: B12.hs

```

1 f :: Double -> Double
2 f x = x ** 3 + x
3
4 binarySearch :: Double -> Double -> Double -> Double
5 binarySearch l r n

```

```

6      | abs (f mid - n) <= 0.001 = mid
7      | f mid < n = binarySearch mid r n
8      | otherwise = binarySearch 1 mid n
9      where
10         mid = (1 + r) / 2
11
12 main :: IO ()
13 main = do
14     n <- readLn
15     print $ binarySearch 1 47 n

```

解説.

3.3 しゃくとり法

問題. N 個の整数が黒板に書かれています。書かれている整数は小さい順に A_1, A_2, \dots, A_N です。異なる 2 つの整数のペアを選ぶ方法は全部で $N(N-1)/2$ 通りありますが、その中で差が K 以下であるような選び方は何通りありますか。

Listing 24: A13.hs

```

1 import qualified Data.ByteString.Char8 as BS
2 import Data.ByteString ( ByteString )
3 import Data.Maybe ( fromJust )
4 import Data.Array ( Array, listArray, (!) )
5
6 readInt :: ByteString -> Int
7 readInt = fst . fromJust . BS.readInt
8
9 step :: Int -> Int -> (Int, Int) -> Array Int Int -> Int -> (Int,
10     Int)
11 step l r (start, sum_) arr k
12     | start < r && (arr ! (start + 1) - arr ! l) <= k = step l r
13     (start + 1, sum_) arr k
14     | otherwise = (start, sum_ + start - 1)
15
16 syakutori :: Int -> Int -> Int -> Array Int Int -> Int -> Int ->
17     Int
18 syakutori l r start arr k sum_
19     | l == r = sum_
20     | otherwise = syakutori (l + 1) r newStart arr k newSum
21     where
22         (newStart, newSum) = step l r (start, sum_) arr k
23
24 main :: IO ()
25 main = do
26     [n, k] <- map readInt . BS.words <$> BS.getLine
27     as <- map readInt . BS.words <$> BS.getLine

```

```

25     let left = 1
26         right = n
27         arr = listArray (left, right) as
28         sum_ = 0
29     print $ syakutori left right left arr k sum_

```

解説.

問題. KYOPRO 商店には N 個の品物が売られており、 i 番目の品物は A_i 円です。連続する番号の品物を買う方法は全部で $N(N+1)/2$ 通りありますが、この中で合計価格が K 円いないとなるような買い方は何通りでしょうか。計算量 $O(N)$ で求めてください。

Listing 25: B13.hs

```

1 import qualified Data.ByteString.Char8 as BS
2 import Data.ByteString ( ByteString )
3 import Data.Maybe ( fromJust )
4 import Data.Array ( Array, listArray, (!) )
5
6 readInt :: ByteString -> Int
7 readInt = fst . fromJust . BS.readInt
8
9 step :: Int -> Int -> Int -> Array Int Int -> Int -> Int
10 step l r start arr k
11     | start < r && arr ! (start + 1) - arr ! (l - 1) <= k = step
12         l r (start + 1) arr k
13     | otherwise = start
14
15 syakutori :: Int -> Int -> Int -> Array Int Int -> Int -> Int
16 syakutori l r start arr k
17     | l == r = next - l + 1
18     | otherwise = (next - l + 1) + syakutori (l + 1) r next arr k
19     where
20         next = step l r start arr k
21
22 main :: IO ()
23 main = do
24     [n, k] <- map readInt . BS.words <$> BS.getLine
25     as <- map readInt . BS.words <$> BS.getLine
26     let arr = listArray (0, n) $ scanl (+) 0 as
27     print $ syakutori 1 n 0 arr k

```

解説.

3.4 半分全列挙

問題. $A \cdot B \cdot C \cdot D$ の 4 つの箱があります。各箱には、以下の N 枚のカードが入っています。

- 箱 A には整数 A_1, A_2, \dots, A_N が書かれたカードがある。

- 箱 B には整数 B_1, B_2, \dots, B_N が書かれたカードがある。
- 箱 C には整数 C_1, C_2, \dots, C_N が書かれたカードがある。
- 箱 D には整数 D_1, D_2, \dots, D_N が書かれたカードがある。

あなたはそれぞれの箱から 1 枚ずつカードを取り出す。取り出した 4 枚のカードに書かれた整数の合計が K となる可能性はあるか、判定してください。

Listing 26: A14.hs

```

1 import qualified Data.ByteString.Char8 as BS
2 import Data.ByteString ( ByteString )
3 import Data.Maybe ( fromJust )
4 import Control.Monad ( replicateM )
5 import Data.List ( sort )
6 import Data.Array ( Array, listArray, (!) )
7
8 binarySearch :: Int -> Int -> Array Int Int -> Int -> Bool
9 binarySearch l r arr x
10     | l > r = False
11     | x == arr ! mid = True
12     | x < arr ! mid = binarySearch l (mid - 1) arr x
13     | otherwise = binarySearch (mid + 1) r arr x
14     where
15         mid = (l + r) `div` 2
16
17 main :: IO ()
18 main = do
19     [n, k] <- map readInt . BS.words <$> BS.getLine
20     [as, bs, cs, ds] <- map (map readInt . BS.words) <$>
        replicateM 4 BS.getLine
21     let ps = (+) <$> as <*> bs
22         qs = listArray (1, n * n) . sort $ (+) <$> cs <*> ds
23     putStrLn $ if any (binarySearch 1 (n * n) qs . (k -)) ps then
        "Yes" else "No"
24
25 readInt :: ByteString -> Int
26 readInt = fst . fromJust . BS.readInt

```

解説.

問題. N 枚のカードがあり、 i 枚目 ($1 \leq i \leq N$) のカードには整数 A_i が書かれています。0 枚以上のカードを選ぶ方法は全部で 2^N 通りありますが、選んだカードの合計がちょうど K となるようにする方法は存在しますか。 $N \leq 30$ を満たす入力で 1 秒以内に実行が終わることが望ましいです。

Listing 27: B14.hs

```

1 import Data.List ( subsequences, sort )

```

```

2 import Data.Array ( Array, listArray, (!))
3
4 divList :: Int -> [Int] -> ([Int], [Int])
5 divList len = splitAt (len `div` 2)
6
7 possibleSums :: [Int] -> [Int]
8 possibleSums xs = map sum (subsequences xs)
9
10 binarySearch :: Int -> Int -> Array Int Int -> Int -> Bool
11 binarySearch l r arr x
12     | l > r = False
13     | x == arr ! mid = True
14     | x < arr ! mid = binarySearch l (mid - 1) arr x
15     | otherwise = binarySearch (mid + 1) r arr x
16     where
17         mid = (l + r) `div` 2
18
19 main :: IO ()
20 main = do
21     [n, k] <- map read . words <$> getLine
22     as <- map read . words <$> getLine
23     let (firstHalf, secondHalf) = divList n as
24         firstSums = possibleSums firstHalf
25         secondSums = listArray (1, 2 ^ (n - (n `div` 2))) (sort $
26             possibleSums secondHalf)
27     putStrLn $ if any (binarySearch 1 (2 ^ (n - (n `div` 2)))
28         secondSums . (k -)) firstSums then "Yes" else "No"

```

解説.

3.5 チャレンジ問題

問題. 配列 $A = [A_1, A_2, \dots, A_N]$ が与えられます。大小関係を崩さないように、配列をできるだけ圧縮してください。ここで圧縮とは、以下の条件をすべて満たす配列 $B = [B_1, B_2, \dots, B_N]$ を求める操作です。なお、このような配列 B は 1 通りに決まります。

- 条件 1 B_1, B_2, \dots, B_N は 1 以上の整数である。
- 条件 2 $A_i < A_j$ であるような組 (i, j) については、 $B_i < B_j$ である。
- 条件 3 $A_i = A_j$ であるような組 (i, j) については、 $B_i = B_j$ である。
- 条件 4 $A_i > A_j$ であるような組 (i, j) については、 $B_i > B_j$ である。
- 条件 5 条件 1~4 を満たす中で、配列 B の最大値をできるだけ小さくする。